
Rješenja iz programskog jezika Logo

I. podskupina (do 6. razreda) - Logo 4.0

Prvi zadatak
STUP

to pravokutnik :a :b

  repeat 2 [fd :b rt 90 fd :a rt 90]

end

to stup :n :a :b

  repeat :n [pravokutnik :a :b fd :b]

end

Drugi zadatak
MEDALJA

to medalja :n :d :k

repeat :n [fd :d rt 180 - :k fd :d lt 180 - (360 / :n + :k)]

end

Treći zadatak
VAL

to poluval :a :n :s

rt :s * 90 / :n repeat :n [fd :a * :s rt 180 / :n * :s] lt :s * (180 + 90 / :n)

end

to valrek :a :n :k :r :s

if :a <= 0 [stop]

poluval :a :n/2 :s

if :a > :r [valrek :a - :k :n :k :r 0-:s]

end

to val :a :n :s :k

valrek :a :n :s :k 1

end

Četvrti zadatak
STAZA
to manji :l1 :l2

  if (last :l1) > (last :l2) then op :l1

  if AND ((last :l1) = (last :l2)) ((first :l1) < (first :l2)) then op :l1

  op :l2

end

to prvi :l

make "ret item 1 :l

for "i 1 count :l [make "ret manji :ret item :i :l]

op :ret

end

to staza :l

pu

while [(count :l)>1][make "p prvi :l  setxy :p pd make "l bm :p :l]

pd

if (count :l)=1 then setxy first :l

end

I. podskupina (do 6. razreda) - MSW Logo

Prvi zadatak
STUP

to pravokutnik :a :b

  repeat 2 [fd :b rt 90 fd :a rt 90]

end

to stup :n :a :b

  repeat :n [pravokutnik :a :b fd :b]

end

Drugi zadatak
MEDALJA

to medalja :n :d :k

repeat :n [fd :d rt 180 - :k fd :d lt 180 - (360 / :n + :k)]

end

Treći zadatak
VAL

to poluval :a :n :s

rt :s * 90 / :n repeat :n [fd :a * :s rt 180 / :n * :s] lt :s * (180 + 90 / :n)

end

to valrek :a :n :k :r :s

if :a <= 0 [stop]

poluval :a :n/2 :s

if :a > :r [valrek :a - :k :n :k :r 0-:s]

end

to val :a :n :s :k

valrek :a :n :s :k 1

end

Četvrti zadatak
STAZA
to manji :l1 :l2

  if (last :l1) > (last :l2) [op :l1]

  if and ((last :l1) = (last :l2)) ((first :l1) < (first :l2)) [op :l1]

  op :l2

end

to prvi :l

make "ret item 1 :l

make "len count :l

for [i 1 :len] [make "ret manji :ret item :i :l]

op :ret

end

to bm :i :l

  make "l1 []

  make "nasao 0

  make "len count :l

  for [j 1 :len] [ifelse and(:nasao=0)(:i=item :j :l)[make "nasao 1][make "l1 lput item :j :l :l1]]

  op :l1

end

to staza :l

  pu

  while [(count :l)>1][make "p prvi :l  setxy first :p last :p pd make "l bm :p :l]

  pd

  if (count :l)=1 [setxy first first :l last first :l]

end

II. podskupina (7. i 8. razredi) - Logo 4.0 i MSW Logo

Prvi zadatak
NEBODER

to mreza :a :b :m :n :kut

repeat :m + 1 [rt :kut pd fd :b bk :b lt :kut pu fd :a / :m]

bk :a + :a / :m repeat :n + 1 [pd fd :a bk :a rt :kut pu fd :b / :n lt :kut]

rt :kut bk :b / :n lt :kut

end

to neboder :a :b :c :m :n :k :kut

mreza :c :a :k :m 90

mreza :c :b :k :n :kut

fd :c lt 180 - :kut

mreza :b :a :n :m 90 - :kut

end

Drugi zadatak
VAL

to poluval :a :n :s

make "rez ycor

rt :s * 90 / :n

repeat round :n / 2 [fd :a * :s rt 180 / :n * :s]

make "rez (ycor - :rez) * :s

repeat :n - round :n / 2 [fd :a * :s rt 180 / :n * :s]

lt :s * (180 + 90 / :n)

op :rez

end

to valrek :a :n :k :r :s

if :r < poluval :a :n/2 :s [valrek :a / :k :n :k :r 0-:s]

end

to val :a :n :k :r

valrek :a :n :k :r 1

end

Treći zadatak
STAZA

to staza :l :d

if number? :l [rt 90 fd :l lt 90]

if list? :l [fd :d repeat count :l [staza first :l :d make "l bf :l] bk :d]

end

Četvrti zadatak
NOGOMET

to brojigraca :n

if :n = 0 [op 1]

op 2 * brojigraca :n - 1

end

to pola :n :a :b

if :n = 1 [fd :b bk :b stop]

make "n :n / 2

fd :b

rt 90 fd :n * :a / 2 lt 90 pola :n :a :b

lt 90 fd :n * :a rt 90 pola :n :a :b

rt 90 fd :n * :a / 2 lt 90 bk :b

end

to nogomet :n :a :b

make "n brojigraca :n

lt 90 pola :n :a :b

end

